
1
Decision/action requested

It is requested to include the proposed changes and corrections into draft PM IRP SS
2
References

S5-046040 (draft TS 32.413 PMIRP_CORBA_SS), SA5#36bis
S5-046139 (JobOverloadManagement CR for PM IRP IS), SA5#37
3
Rationale

There are some changes need to be made in PM IRP SS according to the jobOverloadManagement CR for PMIRP IS. Moreover, there are some typos in the document. 
4
Consequences and implications

N/A.
5
Issues of discussion

(This section should, as necessary, contain additional argumentation).

5.2
Operation parameter mapping
The PMIRP: IS 3GPP TS 32.412 [7] defines semantics of parameters carried in operations across the PMIRP. The following tables indicate the mapping of these parameters, as per operation, to their equivalents defined in this SS.

Table 2: Mapping from IS createMeasurementJob parameters to SS equivalents
	IS Operation parameter
	SS Method parameter
	Qualifier

	iocName
	PMIRPSystem::MOClassNameType  moClass
	M

	iocInstanceList
	PMIRPSystem::MOInstanceListType  moInstanceList
	M

	measurementCategoryList
	PMIRPSystem::MeasurementCategoryListType  measurementCategoryList
	M

	granularityPeriod
	PMIRPSystem::GranularityPeriodType  granularityPeriod
	M

	reportingPeriod
	PMIRPSystem::ReportingPeriodType  reportingPeriod
	M

	startTime
	PMIRPSystem::StartTimeTypeOpt  startTime
	O

	stopTime
	PMIRPSystem::StopTimeTypeOpt  stopTime
	O

	schedule
	PMIRPSystem::ScheduleTypeOpt  schedule
	O

	jobId
	Return value of type PMIRPSystem::JobIdType
	M

	unsupportedList
	PMIRPSystem::JUnsupportedListType unsupportedList
	M

	Status
	ManagedGenericIRPConstDefs::Signal
Exception:
CreateMeasurementJob, ManagedGenericIRPSystem::InvalidParameter, 

HighWorkload
	M

	priority
	PMIRPSystem::JobPriorityTypeOpt priority
	O


Table 5: Mapping from IS resumeMeasurementJob parameters to SS equivalents
	IS Operation parameter
	SS Method parameter
	Qualifier

	jobId
	PMIRPSystem::JobIdType jobId
	M

	Status
	PMIRPSystem::ResultType
Exception:

ResumeMeasurementJob, UnknownJob, JobIsNotSuspended, HighWorkload
	M


A.1
IDL specification (file name "PMIRPSystem.idl")

#ifndef PMIRPSystem_idl

#define PMIRPSystem_idl

#include "ManagedGenericIRPSystem.idl"

#include "NotificationIRPConstDefs.idl"

#include "AlarmIRPConstDefs.idl"

#include "TimeBase.idl"

// This statement must appear after all include statements

#pragma prefix "3gppsa5.org"

/* ## Module: PMIRPSystem

This module contains the specification of all operations of PM IRP Agent.

================================================================

*/

module PMIRPSystem

{

   enum ResultType {OK, Failure};

   typedef   string   MOClassNameType;

   typedef   string   MOInstanceType;

   typedef   sequence<MOInstanceType>   MOInstanceListType;

   typedef   string   MeasurementCategoryType;

   typedef   sequence<MeasurementCategoryType>  MeasurementCategoryListType;

   typedef   unsigned long  GranularityPeriodType; // time period is based on 5 minutes.

   typedef   unsigned long  ReportingPeriodType; // time period is based on 5 minutes.

   typedef   TimeBase::UtcT LocalTimeType;

   union StartTimeTypeOpt switch(boolean)

   {

     case TRUE: LocalTimeType value;

   };

   union StopTimeTypeOpt switch(boolean)

   {

     case TRUE: LocalTimeType value;

   };

   typedef string IntervalTimeType;// the IntervalTimeType from 00:00 to 23:59 of one day

   struct IntervalOfDayType

   {

     IntervalTimeType intervalStartTime;

     IntervalTimeType intervalStopTime;

   };

   typedef sequence<IntervalOfDayType> DailySchedulingType;

   typedef unsigned short DayOfWeekType;

   //The value of DayOfWeekType is from 0 to 6.

   //sunday(0),monday(1),tuesday(2),wednesday(3),


   //thursday(4), friday(5), saturday(6)

   typedef sequence<DayOfWeekType> DaysOfWeekType;

   struct WeeklySchedulingElement

   {

     DaysOfWeekType dayOfWeek;

     DailySchedulingType intervalsOfDay;

   };

   typedef sequence<WeeklySchedulingElement> WeeklySchedulingType;

   enum scheduleTypeChoice { Daily, Weekly };

   union ScheduleType switch (scheduleTypeChoice)

   {

     case Daily:  DailySchedulingType dailyScheduling;

     case Weekly: WeeklySchedulingType weeklyScheduling;

   };

   union ScheduleTypeOpt switch(boolean)

   {

     case TRUE: ScheduleType value;

   };

   typedef   unsigned long   JobIdType;

   typedef sequence<JobIdType> JobIdListType;

   struct JUnsupportedType

   {

     MOInstanceType  moInstance;

     MeasurementCategoryType measurementCategory;

     string reason;

   };

   typedef  sequence<JUnsupportedType>  JUnsupportedListType;

   /**

    * Defines the name of an attribute of a Managed Object

    */

   typedef string MOAttributeName; 

   enum  JobStatusType {Scheduled, Active, Suspended, Stoppped};
   enum  JobPriorityType {Low, Medium, High};
   union JobPriorityTypeOpt switch (boolean)

   {

      case TRUE: JobPriorityType value;

   };
   struct JobInfoType

   {

    JobIdType  jobId;

    MOClassNameType  moClass;

    MOInstanceListType  moInstanceList;

    MeasurementCategoryListType  measurementCategoryList;

    GranularityPeriodType  granularityPeriod;

    ReportingPeriodType  reportingPeriod;

    StartTimeTypeOpt  startTime;

    StopTimeTypeOpt  stopTime;

    ScheduleTypeOpt  schedule;

    JobStatusType jobStatus;
    JobPriorityTypeOpt jobPriority;
   };

   typedef sequence<JobInfoType> JobInfoListType;

   typedef string MeasurementTypeNameType;

   typedef string SubCounterNameType;

   typedef string ProbableCauseType;

   typedef string SpecificProblemType;

   typedef any ThresholdValueType;

   enum SeverityType {Warning, Minor, Major, Critical};

   union HysteresisType switch(boolean)

   {

     case TRUE:  long longValue;

     case FALSE: float floatValue;

   };

   enum DirectionType { Increasing, Decreasing};

   struct ThresholdPackElementType

   {

     ThresholdValueType thresholdValue;

     SeverityType severity; // the value shall be  

     // one of Warning, Minor, Major or Critical.

     HysteresisType hysteresis;

   };

   typedef sequence<ThresholdPackElementType> ThresholdPackType;

   struct ThresholdInfoType

   {

     MeasurementTypeNameType measurementTypeName;

     SubCounterNameType subCounterName;

     ProbableCauseType probableCause;

     SpecificProblemType specificProblem;

     DirectionType direction;

     ThresholdPackType thresholdPack;

   };

   typedef sequence<ThresholdInfoType> ThresholdInfoListType;

   typedef GranularityPeriodType MonitorGranularityPeriodType;// time period is based on 5 minutes.

   typedef unsigned long MonitorIdType;

   struct MUnsupportedType

   {

     MOInstanceType  moInstance;

     MeasurementTypeNameType measurementTypeName;

     SubCounterNameType subCounterName;

     string reason;

   };

   typedef  sequence<MUnsupportedType>  MUnsupportedListType;

   enum  MonitorStatusType {MSuspended, MActive};

   typedef sequence<MonitorIdType> MonitorIdListType;

   struct MonitorInfoType

   {

    MonitorIdType  monitorId;

    MOClassNameType  moClass;

    MOInstanceListType  moInstanceList;

    MonitorGranularityPeriodType monitorGranularityPeriod;

    ThresholdInfoListType thresholdInfoList;

   };

   typedef sequence<MonitorInfoType> MonitorInfoListType;

   const string ET_MEASUREMENT_JOB_STATUS_CHANGED = "notifyMeasurementJobStatusChanged";

   const string ET_THRESHOLD_MONITOR_STATUS_CHANGED = "notifyThresholdMonitorStatusChanged";

   /**

   * This block identifies attributes which are included as part of the

   * PMIRP. These attribute values should not 

   * clash with those defined for the attributes of notification 

   * header (see IDL of Notification IRP).

   */

   module AttributeNameValue

   {

    const string JOB_ID = "JOB_ID";

    const string JOB_STATUS = "JOB_STATUS";

    const string REASON = "REASON";

    const string MONITOR_ID = "MONITOR_ID";

    const string MONITOR_STATUS = "MONITOR_STATUS";

   };

   /**

   * the reason specifies whether EM or NE is high workload. The value shall be one

   * of following: emCpuBusy; emHDShortage, emLowMemory, {neCpuBusy, neObjectInstList},

   * {neHDShortage neObjectInstList}, {neLowMemory, neObjectInstList}, maxJobReached, 

   * otherReason.

   */

   exception HighWorkload { string reason; };

   exception UnknownJob { string reason; }; 

   exception JobCannotBeStopped { string reason; }; 

   exception JobAlreadySuspended { string reason; };

   exception JobIsNotSuspended { string reason; };

   exception UnknownThresholdMonitor { string reason; }; 

   exception ThresholdMonitorAlreadySuspended { string reason; };

   exception ThresholdMonitorIsNotSuspended { string reason; };

   /**

   * System fails to complete the operation.  System can provide reason

   * to qualify the exception.  The semantics carried in reason

   * is outside the scope of this IRP.

   */

   exception GetPMIRPVersions { string reason; };

   exception GetPMIRPOperationsProfile { string reason; };

   exception GetPMIRPNotificationProfile { string reason; };

   exception CreateMeasurementJob { string reason; };

   exception StopMeasurementJob { string reason; };

   exception SuspendMeasurementJob { string reason; };

   exception ResumeMeasurementJob { string reason; };

   exception ListMeasurementJobs { string reason; };

   exception CreateThresholdMonitor { string reason; };

   exception DeleteThresholdMonitor { string reason; };

   exception ListThresholdMonitors { string reason; };

   exception SuspendThresholdMonitor { string reason; };

   exception ResumeThresholdMonitor { string reason; };

   interface PMIRP

   {

      readonly attribute string iRPId;

      /**

      * Return the list of all supported PM IRP versions.

      */

      ManagedGenericIRPConstDefs::VersionNumberSet get_PM_IRP_versions (

      )

      raises (GetPMIRPVersions);

      /**

      * Return the list of all supported operations and their supported

      * parameters for a specific PM IRP version.

      */

      ManagedGenericIRPConstDefs::MethodList get_PM_IRP_operations_profile (

         in ManagedGenericIRPConstDefs::VersionNumber pm_irp_version

      )

      raises (GetPMIRPOperationsProfile,

              ManagedGenericIRPSystem::OperationNotSupported,

              ManagedGenericIRPSystem::InvalidParameter);

      /**

      * Return the list of all supported notifications and their supported

      * parameters for a specific PM IRP version.

      */

      ManagedGenericIRPConstDefs::MethodList get_PM_IRP_notification_profile

      (

         in ManagedGenericIRPConstDefs::VersionNumber pm_irp_version

      )

      raises (GetPMIRPNotificationProfile,

              ManagedGenericIRPSystem::OperationNotSupported,

              ManagedGenericIRPSystem::InvalidParameter);

      /**

      * Request to create a MeasurementJob through Itf-N.

      */

      ManagedGenericIRPConstDefs::Signal create_measurement_job (

        in MOClassNameType  moClass,

        in MOInstanceListType  moInstanceList,

        in MeasurementCategoryListType  measurementCategoryList,

        in GranularityPeriodType  granularityPeriod,

        in ReportingPeriodType  reportingPeriod,

        in StartTimeTypeOpt  startTime,

        in StopTimeTypeOpt  stopTime,

        in ScheduleTypeOpt  schedule,
        in JobPriorityTypeOpt  priority,
        out JobIdType jobId,

        out JUnsupportedListType unsupportedList

      )

      raises (CreateMeasurementJob,

              ManagedGenericIRPSystem::InvalidParameter,

              HighWorkload);

      /**

      * Request to stop a MeasurementJob through Itf-N, after which, 

      * the MeasurementJob will still be visible via Itf-N. Whether 

      * the MeasurementJob is thoroughly removed immediately from 

      * the managed system is vendor specific.

      */

      ResultType stop_measurement_job (

        in JobIdType jobId)

      raises (StopMeasurementJob,

              UnknownJob,

              JobCannotBeStopped);

      /**

      * Request to suspend a MeasurementJob 

      */

      ResultType suspend_measurement_job (

        in JobIdType jobId)

      raises (SuspendMeasurementJob,

              UnknownJob,

              JobAlreadySuspended);

      /**

      * Request to resume a MeasurementJob 

      */

      ResultType resume_measurement_job (

        in JobIdType jobId)

      raises (ResumeMeasurementJob,

              UnknownJob,

              JobIsNotSuspended,

              HighWorkload);

     /**

      * Request to list the information of all or of specified 

      * MeasurementJobs

      */

      ResultType list_measurement_jobs (

        in JobIdListType jobListId,

        out JobInfoListType jobInfoList)

      raises (ListMeasurementJobs,

              ManagedGenericIRPSystem::InvalidParameter);

      /**

      * Request to create a ThresholdMonitor to define the threshold 

      * for some specific measurementTypes or subCounters

      */

      ManagedGenericIRPConstDefs::Signal create_threshold_monitor (

        in MOClassNameType  moClass,

        in MOInstanceListType  moInstanceList,

        in ThresholdInfoListType thresholdInfoList,

        in MonitorGranularityPeriodType monitorGranularityPeriod,

        out MonitorIdType monitorId,

        out MUnsupportedListType unsupportedList)

      raises (CreateThresholdMonitor,

              ManagedGenericIRPSystem::InvalidParameter);

      /**

      * Request to delete a specified ThresholdMonitor 

      */

      ResultType delete_threshold_monitor (


    in MonitorIdType monitorId)

      raises (DeleteThresholdMonitor,

              UnknownThresholdMonitor);

      /**

      * Request to list detailed information about all or 

      * specified ThresholdMonitors

      */

      ResultType list_threshold_monitors (

        in MonitorIdListType monitorIdList,

        out MonitorInfoListType monitorInfoList)

      raises (ListThresholdMonitors,

              ManagedGenericIRPSystem::InvalidParameter);

      /**

      * Request to suspend a ThresholdMonitor 

      */

      ResultType suspend_threshold_monitor (

        in MonitorIdType monitorId)

      raises (SuspendThresholdMonitor,

              UnknownThresholdMonitor,

              ThresholdMonitorAlreadySuspended);

      /**

      * Request to resume a ThresholdMonitor 

      */

      ResultType resume_threshold_monitor (

        in MonitorIdType monitorId)

      raises (ResumeThresholdMonitor,

              UnknownThresholdMonitor,

              ThresholdMonitorIsNotSuspended);

   };

   /**

   * Constant definitions for the notifyMeasurementJobStatusChanged notification

   */      

   module notifyMeasurementJobStatusChanged

   {

     const string MANAGED_OBJECT_CLASS = 

       NotificationIRPConstDefs::AttributeNameValue::MANAGED_OBJECT_CLASS;

     const string MANAGED_OBJECT_INSTANCE = 

       NotificationIRPConstDefs::AttributeNameValue::MANAGED_OBJECT_INSTANCE;

     const string NOTIFICATION_ID = 

       NotificationIRPConstDefs::AttributeNameValue::NOTIFICATION_ID;

     const string EVENT_TIME = 

       NotificationIRPConstDefs::AttributeNameValue::EVENT_TIME;

     const string SYSTEM_DN = 

       NotificationIRPConstDefs::AttributeNameValue::SYSTEM_DN;

     const string EVENT_TYPE = ET_MEASUREMENT_JOB_STATUS_CHANGED;

     /**

     * This constant defines the name of the jobId property, 

     * which is transported in the filterable_body fields.

     * The data type for the value of this property

     * is JobIdType.

     */             

     const string JOB_ID = AttributeNameValue::JOB_ID;

     /**

     * This constant defines the name of the jobStatus property, 

     * which is transported in the filterable_body fields.

     * The data type for the value of this property

     * is JobStatusType.

     */             

     const string JOB_STATUS = AttributeNameValue::JOB_STATUS;

     /**

     * This constant defines the name of the reason property, 

     * which is transported in the filterable_body fields.

     * The data type for the value of this property is string.

     */             

     const string REASON = AttributeNameValue::REASON;

   };

   /**

   * Constant definitions for the notifyThresholdMonitorStatusChanged notification

   */      

   module notifyThresholdMonitorStatusChanged

   {

     const string MANAGED_OBJECT_CLASS = 

       NotificationIRPConstDefs::AttributeNameValue::MANAGED_OBJECT_CLASS;

     const string MANAGED_OBJECT_INSTANCE = 

       NotificationIRPConstDefs::AttributeNameValue::MANAGED_OBJECT_INSTANCE;

     const string NOTIFICATION_ID = 

       NotificationIRPConstDefs::AttributeNameValue::NOTIFICATION_ID;

     const string EVENT_TIME = 

       NotificationIRPConstDefs::AttributeNameValue::EVENT_TIME;

     const string SYSTEM_DN = 

       NotificationIRPConstDefs::AttributeNameValue::SYSTEM_DN;

     const string EVENT_TYPE = ET_MEASUREMENT_JOB_STATUS_CHANGED;

     /**

     * This constant defines the name of the monitorId property, 

     * which is transported in the filterable_body fields.

     * The data type for the value of this property

     * is MonitorIdType.

     */             

     const string MONITOR_ID = AttributeNameValue:: MONITOR_ID;

     /**

     * This constant defines the name of the monitorStatus property, 

     * which is transported in the filterable_body fields.

     * The data type for the value of this property

     * is MonitorStatusType.

     */             

     const string MONITOR_STATUS = AttributeNameValue:: MONITOR_STATUS;

     /**

     * This constant defines the name of the reason property, 

     * which is transported in the filterable_body fields.

     * The data type for the value of this property is string.

     */             

     const string REASON = AttributeNameValue::REASON;

   };

};

#endif

